Bewegungen des Erde Mond Systems

Einleitung

Die Bewegung des Mondes um die Erde wird mit diesem MAPLE file in einigen Varianten durchgerechnet und kann untersucht werden.
Die Erde wird als spärischer Körper mit einem Einheitsradius von 1 angenommen.
Der Mond wird als Massepunkt angenommen im Abstand zum Erdmittelpunkt von 60 Erdradien.
Das gemeinschalftliche Gravitations-Nullpotential im Abstand von etwa 0.73 Erdradien wird vernachlässigt.
Die Sonnenbahn wird durch die Neigung der Erdbahn von 23.45 Grad angenommen.
Der Mond besitz eine Inkliniation von 5.15 Grad relativ zur Ekliptik, von 29 Grad in Relation des Erdäquators.

Berechnungen im 2 Koordinaten System

Die Systeme

INERTIAL SYSTEM
: Zentrum ist der Mittelpunkt der Erde (Ursprungsort Erde): O_E
1. Achse senkrecht auf die Zeichenebene, positiv zum Leser
2. Achse links nach rechts "x-Achse"
3. Achse vom Süd- zum Nordpol

ORTS SYSTEM
Zentrum ist der Beobachter (O_B
Bewegt sich mit der Erde. Ist ein Beobachter an seinem Ort (Ursprungsort Beobachter)

Der Ursprung

Dieser ist zur Startzeit "0" auf einem angenommenen Meridian (Kreis der Figur 1). Der Beobachter steht im Ursprung "0_B"

```
> restart;with(plottools):
```

Ein schön Bildchen malen

```
> c:=circle([0,0],1):
```
Erde

```
> EQ := line([-1,0], [1, 0], color=black):
```
Äquator

```
> l[ecliptic]:=line([-1.4*cos(Pi/180*23.45),1.4*sin(Pi/180*23.45)],
[1.4*cos(Pi/180*23.45),-1.4*sin(Pi/180*23.45)],color=red,
linestyle=1):
```
Ekliptik

```
> l[moon]:=line([-1.4*cos(Pi/180*28.5),1.4*sin(Pi/180*28.5)]
, [1.4*cos(Pi/180*28.5),-1.4*sin(Pi/180*28.5)],
color=blue, linestyle=3):
```
Ebene der Mondbahn

```
> O_E := circle([0,0],0.02,color=blue):
```
Mittelpunkt der Erder (Ursprung Erde)

```
> O_M := circle([cos(Pi/6), sin(Pi/6)],0.02,color=red):
```
Standort des Beobachters; Ursprung des Orts Systems

```
> lR:=arrow([0,0],vector([cos(Pi/6),sin(Pi/6)]),.005,.02,.1,
color=black):
```
Erd Radius

```
> l[Omega]:=arrow([0,1],vector([0,
0.4]),.005,.03,.2,color=black):
```
Vektor der Winkelgeschwindigkeit

```
> l2:= arrow([1,0],vector([0.4,
0]),.005,.03,.08,color=black):
```
Die 2. Achse: Meridian Tangente

```
> l3:= arrow([0, 0],vector([0, .5]), .005, .02, .1,
color=black):
```
Die 3. Achse: Senkrechte zur Erdoberfläche

```
> l_2:= arrow([cos(Pi/6), sin(Pi/6)], vector([1/3*sin(Pi/6),
-1/3*cos(Pi/6)]), .005, .02, .08, color=black):
```
Die Tangente auf dem Meridian

```
> l_3:= arrow([cos(Pi/6), sin(Pi/6)], vector([1/3*cos(Pi/6),
1/3*sin(Pi/6)]), .005, .02, .08, color=black):
```
Die Senkrechte auf der Erdoberfläche

```
> with(plots):
Warning, the names arrow and changecoords have been redefined
```

```
> t0:=textplot([-1,1.2,`Bild 1`],font=[COURIER,BOLD,10]):
> t1:=textplot([.3,.3,`R=1`]):
> t2:=textplot([.28,.07,`theta`]):
> t3:=textplot([1.3,0.75,`3.-Achse im Winkel Theta`]):
> t4:=textplot([1.2,.3,`2.-Achse`]):
> t5:=textplot([0.1,0.6,`3.-Achse `]):
> t6:=textplot([-0.2,1.3,`Omega`]):
> t7:=textplot([-0.5,0.1,`Äquator`]):
> t8:=textplot([1.3,.1,`Achse des Horizonts`]):
> t9:=textplot([1.1,-.37,`Ekliptik`],color=red):
> t11:=textplot([-0.01,-.08,`O-Erde`],color=blue):
> t10:=textplot([-.5,.5,`Ebene der Mondbahn`],color=blue):
> t12:=textplot([1.2,0.47,`O-Beobachter`],color=red):
```

```
> x1:=(r,phi)->r*cos(phi):x2:=(r,phi)->r*sin(phi):
```
Mit Hilfe von polaren Koordinaten

```
> p:=plot([x1(.4,phi),x2(.4,phi),phi=0..Pi/6]):
> display({EQ,O_E,O_M,c,lR,l[Omega],l[moon],l[ecliptic],l2,l
3,l_2,l_3,p,t0,t1,t2,t3,t4,t5,t6,t8,t7,t9,t10,t11,t12},sca
ling=constrained,axes=none);
```

Bild 1



Koordinaten Transformation des rotierenden Systems

```
> restart;with(LinearAlgebra):
```

Im Trägheits System haben die Achsen ihre zugewiesenen Richtungen = Einheitsvektoren

```
> e10:=Vector[row]([1,0,0]);e20:=Vector[row]([0,1,0]);e30:=Vector[row]([0,0,1]);
```

$$e10 := [1, 0, 0]$$
$$e20 := [0, 1, 0]$$
$$e30 := [0, 0, 1]$$

Schritt 1:

Das System rotiert. Dir benutzen die Winkeleinhaiten radian/Tag. Nach einer beliebigen Rotation sind die Achsenrichtungen neu zu bestimmen.

Zuerst bestimmen wir die Vektoren ...

```
> e11:=Vector([cos(phi),cos(Pi/2+phi),cos(Pi/2)]):
> e21:=Vector([cos(Pi/2-phi),cos(phi),cos(Pi/2)]):
> e31:=Vector([0,0,1]):
```

... dann wird daraus die Matrix berechnet.

```
> M1:=Matrix([e11,e21,e31]);
```

$$M1 := \begin{bmatrix} \cos(\phi) & \sin(\phi) & 0 \\ -\sin(\phi) & \cos(\phi) & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

Schritt 2:

Das System rotiert um um die aus der Zeichenebne kommenden Achse.

Wie vorher: erst einmal Aufbereitung der Vektoren

```
> e12:=Vector([1,0,0]):
> e22:=Vector([0,cos(Pi/2-theta),cos(theta)]):
> e32:=Vector([0,cos(Pi-theta),cos(Pi/2-theta)]):
```

... dann Zusammenbau der Drehmatrix

```
> M2:=Matrix([e12,e22,e32]);
```

$$M2 := \begin{bmatrix} 1 & 0 & 0 \\ 0 & \sin(\theta) & -\cos(\theta) \\ 0 & \cos(\theta) & \sin(\theta) \end{bmatrix}$$

Beide Transformationen werden miteinander verbunden (kombiniert).

Dazu verwenden wir die transponierte Form beider Drehmatrizen:

```
> M:=Transpose(M1).Transpose(M2);
> MT:=Transpose(M);
```

$$M := \begin{bmatrix} \cos(\phi) & -\sin(\phi)\sin(\theta) & -\sin(\phi)\cos(\theta) \\ \sin(\phi) & \cos(\phi)\sin(\theta) & \cos(\phi)\cos(\theta) \\ 0 & -\cos(\theta) & \sin(\theta) \end{bmatrix}$$

$$MT := \begin{bmatrix} \cos(\phi) & \sin(\phi) & 0 \\ -\sin(\phi)\sin(\theta) & \cos(\phi)\sin(\theta) & -\cos(\theta) \\ -\sin(\phi)\cos(\theta) & \cos(\phi)\cos(\theta) & \sin(\theta) \end{bmatrix}$$

Schritt 3: Überprüfung.

Wenn wir alles richtig gemacht haben, sollten wir durch die Multiplikation der Matrix mit ihrer Transponierten die Einheitsmatrix rausbekommen

.

```
> simplify(MT.M);
```

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

```
> with(plots):with(plottools):
Warning, the name changecoords has been redefined

Warning, the name arrow has been redefined
```

Wir schauen uns das Vektorsystem "Erdmittelpunkt - Beobachter - dessen Sicht zum Mond" an

```
> Erde:=circle([0,0],1):
> RA :=arrow([0,0],[cos(Pi/4),sin(Pi/4)],.02,.08,.1,
  color=green):
> X:=arrow([0,0],vector([1/2*2^(1/2)+6, 1/2*2^(1/2)+2]),
  .02, .08, .05, color=red):
> Z :=arrow([cos(Pi/4),sin(Pi/4)], vector([6, 2]), .02, .08,
  .05, color=red):
> ORI := circle([0,0],0.04,color=blue):
```

```
> O_RI := circle([cos(Pi/4), sin(Pi/4)],0.04,color=red):
> Moon := circle([6.7,2.7],0.1,color=blue):
> T0:=textplot([1.6,2.6,`Figure2`],font=[COURIER,BOLD,10]):
> T1:=textplot([3.26,1.8,`z(t)`]):
> T2:=textplot([0.23,0.5,`R(t)`]):
> T3:=textplot([3.9,1.2,`x(t)`]):
> T4:=textplot([-0.01,-0.15,`O = Erdmittelpunkt`]):
> T5:=textplot([0.6,0.9,`Beobachter = O-B`]):
> T6:=textplot([6.7,3.0,`Mond`]):
> display({Erde,X,RA,Z,T0,T1,T2,T3,T4,T5,T6,ORI,O_RI,Moon});
```



Wie der Mond die Erde umrundet

Die Erde hat die ca. 81.3 fache Masse des Mondes. Der Mond beschreibt in erster Näherung eine Bahn um die Erde von 60 Erdradien.
Wir berechnen das System mit den oben genannten Annahmen:

```
> r:=60:omega[m]:=Pi/14:alpha:=Pi/6:
> y:=Vector[row]([r*cos(omega[m]*t),r*sin(omega[m]*t),0]);
```

$$y := \left[60\cos\left(\frac{\pi\, t}{14}\right), 60\sin\left(\frac{\pi\, t}{14}\right), 0 \right]$$

Die Koordinaten Richtungen der geneigten Mondbahn (Inklination) werden in den folgenden Einheitsvektoren eingetragen:

```
> e1:=Vector([1,0,0]):
> e2:=Vector([0,cos(alpha),cos(Pi/2-alpha)]):
> e3:=Vector([0,cos(Pi/2+alpha),cos(alpha)]):
```

Wie bereits kennengelernt bilden wir aus diesen Vektoren die Drehmatrix

```
> Mt:=Matrix([e1,e2,e3]);
```

$$Mt := \begin{bmatrix} 1 & 0 & 0 \\ 0 & \frac{\sqrt{3}}{2} & \frac{-1}{2} \\ 0 & \frac{1}{2} & \frac{\sqrt{3}}{2} \end{bmatrix}$$

Die Bahn als Funktion des Weges x(t) in den Koordinaten des Trägheitssystems ist. Multiplikation der y-Komponente der bahn mit der Drehmatrix ergibt die zugehörige x-Komponente

```
> y.Mt;
```

$$\left[60\cos\left(\frac{\pi\, t}{14}\right), 30\sin\left(\frac{\pi\, t}{14}\right)\sqrt{3}, -30\sin\left(\frac{\pi\, t}{14}\right) \right]$$

```
> x:=Transpose(%);
```

$$x := \begin{bmatrix} 60\cos\left(\frac{\pi\, t}{14}\right) \\ 30\sin\left(\frac{\pi\, t}{14}\right)\sqrt{3} \\ -30\sin\left(\frac{\pi\, t}{14}\right) \end{bmatrix}$$

Wir können daraus folgende Zeitabhängigkeit formulieren. Das passiert so:
Zunächst wird der vorher berechnete Vektor als Gesamteinheit "aufgelöst = unapply.
Dann werden den Einzelkomponenten des Vektors dessen Zeitbeziehung mitgegeben. Durch das ,t in den folgenden Zeilen.

```
> x1:=unapply(x[1],t);
> y1:=unapply(x[2],t);
> z1:=unapply(x[3],t);
```

$$x1 := t \to 60\cos\left(\frac{1}{14}\pi\, t\right)$$

$$y1 := t \to 30\sin\left(\frac{1}{14}\pi\, t\right)\sqrt{3}$$

$$z1 := t \to -30\sin\left(\frac{1}{14}\pi\, t\right)$$

Damit ist die zeitliche Bewegung der Mondbahn bestimmt. Wir schauen uns diese aus einer Distanz aus dem Weltraum aus an:

```
> S:=spacecurve([x1(t),y1(t),z1(t)],t=0..28,color=blue,labels=[
`x-Achse`,`y-Achse`,`z-Achse`]):
> earth:=sphere([0,0,0],6):
> K2:=seq(sphere([x1(i),y1(i),z1(i)],4),i=1..28):
> moon:=display(K2,insequence=true):
> tp:=textplot([1.6,2.6,`Bild 3`],font=[COURIER,BOLD,10]):
> display({S,earth,moon},scaling=unconstrained,style=patchnogri
d,axes=framed,title=`Bild 3 : Animation Mond umkreist die
Erde (im Trägheistssytem`);
```

Bild 3 : Animation Mond umkreist die Erde (im Trägheistssytem



Das Orts System der Erde

Werden die beiden Vektoren x und R mit der Matrix M multipliziert, dann ergibt sich gemäß der

ersten Gleichung:
Die Bewegung des Mondes im Orts System

```
> phi:=2*Pi*t:
```
Erst einmal der Winkel

```
> R:=Vector([0,cos(theta),sin(theta)]);
```
Dann der Abstand

$$R := \begin{bmatrix} 0 \\ \cos(\theta) \\ \sin(\theta) \end{bmatrix}$$

```
> xt:=MT.x;Rt:=MT.R;
```

$$xt := \begin{bmatrix} 60\cos(\phi)\cos\left(\frac{\pi t}{14}\right) + 30\sin(\phi)\sin\left(\frac{\pi t}{14}\right)\sqrt{3} \\ -60\sin(\phi)\sin(\theta)\cos\left(\frac{\pi t}{14}\right) + 30\cos(\phi)\sin(\theta)\sin\left(\frac{\pi t}{14}\right)\sqrt{3} + 30\cos(\theta)\sin\left(\frac{\pi t}{14}\right) \\ -60\sin(\phi)\cos(\theta)\cos\left(\frac{\pi t}{14}\right) + 30\cos(\phi)\cos(\theta)\sin\left(\frac{\pi t}{14}\right)\sqrt{3} - 30\sin(\theta)\sin\left(\frac{\pi t}{14}\right) \end{bmatrix}$$

$$Rt := \begin{bmatrix} \sin(\phi)\cos(\theta) \\ \cos(\phi)\sin(\theta)\cos(\theta) - \cos(\theta)\sin(\theta) \\ \cos(\phi)\cos(\theta)^2 + \sin(\theta)^2 \end{bmatrix}$$

z(t) wird erhalten durch die Subtraktion der vektoren xt und Rt

```
> zt:=xt-Rt:
```
gemäß Bild 2

from which we extract the three coordinates (as functions of time *t* and receive the equations, which comprise our perceptions as observers in the **local system** :

```
> z[1]:=simplify(unapply(zt[1],t));
> z[2]:=simplify(unapply(zt[2],t));
> z[3]:=simplify(unapply(zt[3],t));
```

$$z_1 := t \to 60\cos(2\pi t)\cos\left(\frac{1}{14}\pi t\right) + 30\sin(2\pi t)\sin\left(\frac{1}{14}\pi t\right)\sqrt{3} - \sin(2\pi t)\cos(\theta)$$

$$z_2 := t \to -60\sin(2\pi t)\sin(\theta)\cos\left(\frac{1}{14}\pi t\right) + 30\cos(2\pi t)\sin(\theta)\sin\left(\frac{1}{14}\pi t\right)\sqrt{3} + 30\cos(\theta)\sin\left(\frac{1}{14}\pi t\right) - \cos(2\pi t)\sin(\theta)\cos(\theta) + \cos(\theta)\sin(\theta)$$

$$z_3 := t \to -60\sin(2\pi t)\cos(\theta)\cos\left(\frac{1}{14}\pi t\right) + 30\cos(2\pi t)\cos(\theta)\sin\left(\frac{1}{14}\pi t\right)\sqrt{3} - 30\sin(\theta)\sin\left(\frac{1}{14}\pi t\right) - \cos(2\pi t)\cos(\theta)^2 - \sin(\theta)^2$$

Der Beobachter steht im Mittelpunkt des Geschehens

Mittels der bisher dargestellten Gleichungen lassen wir jetzt den Mond einmal um den Beobachter kreisen. Die Erde wird weggelassen.
Klar, der Beobachter sieht nur bis zum Horizont!!

Erste Sicht der Dinge

Wir befinden uns in Westeuropa so etwa auf der Höhe des schönen Schwabenlandes. Breite ca. 49 Grad Nord. Das ist etwas Nördlich Stuttgart.

Da die Erde sich vorwärtsbewegt, sihet es so aus, wals würde der Mond sich spiralförmig um den Beobachter drehen. Pro Tag zeigt sich die Mondbewegung als ein
Kreis, dessen Enden immer etwas weiter weggerückt sind.
Diese Bewegung wird im folgenden Bild 4 dargestellt.

```
> theta:= 49 * Pi /180:
> s1:=spacecurve([z[1](t),z[2](t),z[3](t)],t=0..7,numpoints=
  300,color=black,labels=[`z[1]`,`z[2]`,`z[3]`]):
> s2:=spacecurve([z[1](t),z[2](t),z[3](t)],t=21..28,linestyl
  e=3,numpoints=300,color=red):
> Oberver:=sphere([0,0,0],2,color=blue):
> tp:=textplot3d([0,0,6,`Beobachter`],color=black):
> K3:=seq(sphere([z[1](i),z[2](i),z[3](i)],4),i=1..28):
> Moon:=display(K3,insequence=true):
> display({s1,s2,Moon,Oberver,tp},scaling=unconstrained,axes
  =framed,labels=[`z[1]`,`z[2]`,`z[3]`],style=patchnogrid,or
  ientation=[-40,50],title=`Bild 4.1: Die scheinbare
  Mondbewegung über einen Zyklus`);
```

Bild 4.1: Die scheinbare Mondbewegung über einen Zyklus



Zweite Sicht der Dinge

Es wird von der Psotion des Beobachtes im Orts System geschaut. Der Beobachter schaut in Richtung einer der drei Koordinaten. Für das Modell ist es
unerheblich, in welche Koordinatenrichtung geschaut wird. Die Koordinatenrichtungen sind hier, da die Erde aus dem Bild entfernt wurde gleichberechtigt.
Der Beobachter sieht nun die Mondbewwegung als Projektionsspur auf den anderen Koordinaten.

```
> observer:=circle([0,0],1):
> p1:=plot([z[1](t),z[2](t),t=0..7],color=black):
> p2:=plot([z[1](t),z[2](t),t=21..28],color=red):
> display({p1,p2,observer},scaling=constrained,title=`Bild
  4.2:  Die Mondbahn als Projektion auf den
```

```
Koordinatenebenen z1, z2`,labels=[`z[1]`,`z[2]`]);
```

Bild 4.2: Die Mondbahn als Projektion auf den Koordinatenebenen z1, z2




```
> p3:=plot([z[2](t),z[3](t),t=0..7],color=black,labels=[`z[2
  ]`,`z[3]`],numpoints=1500):
> p4:=plot([z[2](t),z[3](t),t=21..28],color=red,labels=[`z[2
  ]`,`z[3]`],numpoints=1500):
> display({p3,p4,observer},scaling=constrained,
> title=`Bild 4.3:  Die Mondbahn als Projektion auf den
  Koordinatenebenen z2, z3`);
```



Zum Schluss: die Mondphasen

Es werden unsere bekannten Gleichungen des Inertialsystems verwendet. Wir berechnen einen Siderischen Monat und visualisieren die Mondphasen:
Neumond, Halbmond aufgehend, Vollmond, Halbmond abnehmend. Im Englisch heiist das übrigens: new moon, waxing moon, full moon, waning moon.

Die Gleichungen der Koordinaten von etwas weiter oben: x1, y2, z1 werden hier angewandt.
Die Neigung der Ekliptik beträgt hierbei 5.15 Grad.

```
>
> S:=spacecurve([x1(t),y1(t),z1(t)],t=0..28,color=blue,labels=[
  `1-axis`,`2-axis`,`3-axis`]):
> earth:=sphere([0,0,0],6):
> K2:=seq(sphere([x1(7*i),y1(7*i),z1(7*i)],4),i=1..4):
```

```
> moon:=display(K2):
> Tp:=textplot3d([[0,-50,20,`Neumond`],[80,0,0,`zunehmend`],[0,
  38,-34,`Vollmond`],[-80,0,0,`abnehmend`],[0,0,-15,`Earth`]],c
  olor=black):
> AR1:=arrow([0,-100,50],vector([0,30,-15]),1,4,.2,color=red):
> AR2:=arrow([-60,-70,35],vector([0,30,-15]),1,4,.2,color=red):
> AR3:=arrow([60,-70,35],vector([0,30,-15]),1,4,.2,color=red):
> display({S,earth,moon,Tp,AR1,AR2,AR3},
> scaling=constrained,axes=framed,style=patchnogrid,title=`Bild
  5: Mondphasen, rote Pfeile sind Sonnenstrahlen`);
```

Bild 5: Mondphasen, rote Pfeile sind Sonnenstrahlen